A MORE INFORMATIVE STATISTICAL ANALYSIS

FOR PREDATOR-PREY STUDIES*

by

David J. Kolar and Jacquelyn F. Sullivan-Kwasnik
Purdue University

Department of Statistics

Division of Mathematical Sciences

Mimeograph Series #490

August, 1977

*This research was supported by the Bionucleonics Department of Purdue University.

INTRODUCTION

Statistical analyses in most predator-prey experiments is limited to determining whether a particular stress or toxicant affects prey vulnerability by either increasing or decreasing the probability that treated organisms will be differentially preyed upon. This paper proposes methods which (1) estimate how much a treatment increases or decreases prey vulnerability and (2) find significant differences between treatments. The additional calculations that are required more than double the information obtained from earlier analyses, yet remain a small fraction of the overall effort.

The typical predator-prey study, utilizing fish as the test organisms, exposes randomly selected members of a prey population to a sublethal level of a stress or toxicant. Following exposure, M members of the treated group and N members of the control group (usually M = N) are placed in each of L experimental predator-prey chambers. After the prey fish have become acclimated to their surroundings a predator is added. When approximately half the prey have been consumed the number T of treated fish eaten and the number C of control fish eaten is recorded for each tank. The above process is repeated for as many treatment levels as the investigator wishes to test.

The predator-prey study on which the following techniques were first applied was undertaken by Jacquelyn F. Sullivan as part of her Ph.D. research for the Bionucleonics Department of Purdue University. This research was conducted in an innovative manner that recommends itself as a model for future predator-prey studies. Primary innovations consisted of utilizing model ecosystems and sustained behavioral observations. Data from this experiment are used in examples throughout this paper; therefore, the essential results are provided in Table 1.

A STANDARD ANALYSIS

Predator-prey experiments provoke two main inquiries. The first inquiry concerns whether a particular treatment or level of treatment influences prey vulnerability. Normally, the investigator assumes that a treatment either will have no effect or will increase prey vulnerability. In such cases a one-sided hypothesis test is appropriate. An investigator, having knowledge of similar studies, may be fairly certain that a given treatment will double prey vulnerability; then a two-sided hypothesis test becomes appropriate. The latter section of this paper contains statistical procedures which test appropriate hypotheses and give point estimates and confidence intervals for the increase in prey vulnerability due to treatment. The second inquiry asks whether the treatments vary in their affect on prey vulnerability. Answering this inquiry is the purpose of this section.

The use of standard techniques to find significant differences between treatments will be illustrated with an example. In the Sullivan experiment, nine treatments were used. Common sense dictated that comparisons of interest be made among three subsets of the nine treatments: (1) acute exposure treatments; (2) chronic exposure treatments; and (3) the 0.05 mg Cd/liter treatments. The values of T/R (R equals the number of prey fish eaten during a subexperiment) for each treatment are listed in Table 2.

TABLE 1

Experimental results, maximum likelihood,estimators of K, Σ_1 and Σ_1' confidence limits for K, and H_o: K=1 versus H_1: K>1 P values. Positions where an asterisk appears indicate that M=N=14 for that subexperiment. Elsewhere, M=N=20. R denotes the number of fish eaten during the course of a subexperiment.

Mg Cd/ liter	Exposure Time	Tank	R	$T_{exp.}$	MLE(K)	90% C.I.(K)	$P(\Sigma_1 \leq \Sigma_{1,exp} \mid K=1)$
.500	48 hr.	1	18	12	2.55	(1.8,6.7)	.000
	48 hr.	2	16	11			
	48 hr.	3	20	13			
.375	48 hr.	1	20	13	3.33	(2.1,6.9)	.000
	48 hr.	2	21	15			
	48 hr.	3	19	14			
.250	48 hr.	1	16	9	.77	(0.4,1.3)	.820
	48 hr	2	19	8			
	48 hr.	3	16	6			
.050	48 hr.	1*	16	8	.49	(.4,.8)	.937
	48 hr.	2*	12	3			
	48 hr.	3	18	6			
.050	21 d.	1	24	14	1.84	(1.3,4.0)	.004
	21 d.	2	20	12			
	21 d.	3	21	13			
.025	21 d.	1	21	15	3.95	(1.7,5.2)	.000
	21 d.	2	20	12			
	21 d.	3	20	17			
.019	21 d.	1	21	11	1.36	(.8,1.9)	.214
	21 d.	2	24	11			
	21 d.	3	20	14			
.013	21 d.	1	20	8	.89	(0.5,1.4)	.614
	21 d.	2	17	11			
	21 d.	3	21	9			
.050 R	21 d.	1	22	13	2.02	(1.4,4.3)	.002
	21 d.	2	22	14			
	21 d.	3	21	13			

TABLE 2

Cd Conc. (mg/liter)	Exposure Time	Tank 1	T/R Tank 2	Tank 3
0.500	48 hr	12/18	11/16	13/20
0.375	48 hr	13/20	15/21	14/19
0.250	48 hr	9/16	8/19	6/16
0.050	48 hr	8/16	3/12	6/18
0.050	21 day	14/24	12/20	13/21
0.025	21 day	15/21	12/20	17/20
0.019	21 day	11/21	11/24	14/20
0.013	21 day	8/20	11/17	9/21
0.050*	21 day	13/22	14/22	13/21

The proportion T/R or a variance stabilizing transformation of T/R is commonly used as the basic statistic for the analysis of variance of binomial type data. Although data from this experiment is not binomially distributed, the value T/R was chosen as the basic statistic for the following reasons. As prey vulnerability increases from treatment to treatment, the $E[T/R|R]$ increases. A conservative estimate for the variance of $(T/R|R)$ can easily be computed for each observation; subsequently, these variances can be used to estimate the residual mean square. Data from studies of this sort tends to become binomial as M and N ($M = N$) are allowed to increase toward infinity while R is held below an upper bound. Finally, assumptions of homogeneity of variance and normality were met sufficiently well to ensure robustness for analysis of variance techniques.

Since T/R given R is discrete, it cannot have a normal distribution. But it is approximately normal for a wide range of prey vulnerability levels K in the sense that a normal approximation using the mean and variance of T/R|R,K yields a good estimate of the actual T/R probability density function (PDF). Table 3 compares the actual PDF of T/R for M = N = 20, R = 20 when K = 1.00 and 0.20. The value of K is 1.00 when a treatment has no effect; K = 0.20 when a treatment decreases prey vulnerability five-fold.

TABLE 3. M = N = R = 20

K = 1.00			K = 0.20		
T/R	PDF	NORM. APP.	T/R	PDF	NORM. APP.
.25	.002	.002	.00	.000	.001
.30	.011	.012	.05	.004	.008
.35	.044	.045	.10	.034	.039
.40	.115	.115	.15	.126	.120
.45	.205	.204	.20	.245	.231
.50	.248	.244	.25	.275	.274
.55	.205	.204	.30	.193	.203
.60	.115	.115	.35	.089	.093
.65	.044	.045	.40	.027	.026
.70	.011	.012	.45	.006	.005
.75	.002	.002	.50	.001	.001

A tabular study of the influence of K and R on the variance of T/R [= P], arc sin P [= ASP] and arc sin (2P-1) [= A2P] is presented in Tables 4 and 5. Table 4 examines the variance stabilizing effect of arcsine transformations when M = N = R. The function B is defined for argument X as follows:

$$B(X) = \frac{[\mathrm{Max}_{K \varepsilon K \text{ interval}}\{\mathrm{Var}(X)|K,M,N,R\}]}{[\mathrm{Min}_{K \varepsilon K \text{ interval}}\{\mathrm{Var}(X)|K,M,N,R\}]}$$

Values of B(X) near one indicate excellent homogeneity of variance. Table 4 indicates a strong variance stabilizing effect for both arcsine transformations as M = N = R increases. Note, however, that variances remain nearly homogeneous as long as treatment effects K stay in a range from 0.20 to 5.00.

TABLE 4

M=N=R	K Interval	E(P) Interval	B(P)	B(ASP)	B(A2P)
5	0.10 - 10.0	.152 - .848	1.53	1.97	1.97
	0.20 - 5.0	.236 - .764	1.24	1.67	1.67
	0.30 - 3.3	.295 - .705	1.13	1.40	1.40
10	0.10 - 10.0	.160 - .840	1.60	1.74	1.74
	0.20 - 5.0	.241 - .759	1.27	1.24	1.24
	0.30 - 3.3	.299 - .701	1.14	1.10	1.10
15	0.10 - 10.0	.162 - .838	1.62	1.38	1.38
	0.20 - 5.0	.243 - .757	1.27	1.12	1.12
	0.30 - 3.3	.300 - .700	1.15	1.06	1.06
20	0.10 - 10.0	.163 - .837	1.63	1.23	1.23
	0.20 - 5.0	.243 - .757	1.28	1.09	1.09
	0.30 - 3.3	.300 - .700	1.15	1.05	1.05
25	0.10 - 10.0	.163 - .837	1.63	1.18	1.18
	0.20 - 5.0	.244 - .756	1.28	1.08	1.08
	0.30 - 3.3	.301 - .699	1.15	1.04	1.04

Table 5 examines the efficacy of arcsine transformations when M = N = 20 for various K intervals and R ranges. Here the function B(X) is defined:

$$B(X) = \frac{\text{Max}_{K\varepsilon K \text{ interval, } R\varepsilon R \text{ values}}\{\text{Var}(X)\,|\,K,M{=}N{=}20,R\}]}{\text{Min}_{K\varepsilon K \text{ interval, } R\varepsilon R \text{ values}}\{\text{Var}(X)\,|\,K,M{=}N{=}20,R\}]}$$

This table indicates a general increase in heterogeneity of variance for P, ASP and A2P as R is permitted to fluctuate about twenty. The arcsine transformations also tend to lose their variance stabilizing effect under these circumstances, at least for the K intervals examined here.

TABLE 5

R Values	K int.	B(P)	B(ASP)	B(A2P)
20	1.00	1.00	1.00	1.00
	.5 - 2.0	1.05	1.02	1.02
	.2 - 5.0	1.28	1.09	1.09
19,20,21	1.00	1.22	1.23	1.23
	.5 - 2.0	1.28	1.25	1.25
	.2 - 5.0	1.58	1.39	1.39
18,...,22	1.00	1.49	1.51	1.51
	.5 - 2.0	1.57	1.55	1.55
	.2 - 5.0	1.97	1.78	1.78
17,...,23	1.00	1.83	1.86	1.86
	.5 - 2.0	1.93	1.93	1.93
	.2 - 5.0	2.50	2.42	2.42
16,...,24	1.00	2.25	2.30	2.30
	.5 - 2.0	2.39	2.42	2.42
	.2 - 5.0	3.24	3.35	3.35

The foregoing study implies that considerable heterogeneity of variance exists between T/R values from predator-prey experiments. In the most extreme case from Table 5, it was possible for variances to differ by a ratio of 3.24 to 1.00. Since analysis of variance techniques remain robust for variance heterogeneity of this size[1], we are justified in proceeding with an analysis of variance on the Sullivan data for maximum likelihood estimators of K (next section) remain between 0.20 and 5.00 and R fluctuations about 20 remained between 16 and 24. The data, including the treatment cell where some tanks started with fourteen control and fourteen treated fish, easily passed Cochran's test for homogeneity of variance (P = .962).

A conservative estimate of the variance of each T/R value may be obtained by assuming that K = 1 (i.e. no treatment effect exists). Then the distribution of T is hypergeometric so that:

$$\begin{aligned} \mathrm{Var}(T/R|K=1,R,M,N) &= \mathrm{Var}(T|K=1,R,M,N)/R^2 \\ &= [MNR(M+N-R)]/[(M+N)^2(M+N-1)R^2] \end{aligned}$$

One may also generate these variances with the program of Appendix A. Variances for the example data are listed in Table 6.

An estimate of the residual mean square is calculated in the following manner:

$$\text{Residual MS} = \Sigma_{i=1}^{10} (F_i V_i)/\Sigma_{i=1}^{10} F_i = 0.006865$$

---

[1]Anderson, p. 17.

Since this is a conservative (i.e. high) estimate of the average variance among observed results, the degrees of freedom for this residual mean square are assumed to be infinite.

TABLE 6

Variance estimates for tank-treatment results of the Sullivan experiment.

i	M=N	R	F (frequency)	V[Var(T/R|K=1,R,M,N)]
1	14	12	1	.012346
2	14	16	1	.006944
3	20	16	3	.009615
4	20	17	1	.008673
5	20	18	2	.007835
6	20	19	2	.007085
7	20	20	7	.006410
8	20	21	6	.005800
9	20	22	2	.005245
10	20	24	2	.004274

The analysis of variance provided in Table 7 assumes that the three tanks as well as the nine treatments may influence the outcome of each subexperiment. As it turns out neither tanks nor the tank-treatment interaction is significant. Treatments, however, are significantly different.

TABLE 7

Analysis of variance for Sullivan experiment. The symbol *** denotes 0.001 significance.

DF	Source	Sum of Squares	Mean Square	F
2	Tank	.005	.0025	0.36
8	Trt	.348	.043	6.34***
16	Tank x Trt	.152	.010	1.38
Infinite	Residual		.006865	

A Newman Keuls multiple comparison test[2] produces the following treatment ranking:

TRT	50A	250A	13Ch	19Ch	50Ch	50Res	500A	375A	25Ch
Mean	.3611	.4529	.4919	.5607	.6008	.6154	.6681	.7004	.7214

This ranking may be subdivided into three meaningful subrankings:

(1) 50A 50Ch 50Res

(2) 50A 250A 500A 375A

(3) 13Ch 19Ch 50Ch 25Ch

Treatments which are connected by the same line are not significantly different.

[2] Anderson, p. 10.

In summary, analysis of variance techniques utilizing T/R as the response variable seem appropriate when M equals N with both greater than nineteen (normality and homogeneity of variance assumptions were examined here chiefly for M=N=20) and when R values are kept near M. The residual mean square for the analysis of variance and multiple comparisons should be calculated in the manner discussed.

SPECIAL ANALYSIS

An important parameter in the following pages is the measure K of increased prey vulnerability due to treatment. If M treated and N control prey inhabit a tank with a predator, it is assumed that:

$$P_{t,1} = P_{t,2} \cdots = P_{t,M} = P_t \quad \text{and}$$

$$P_{c,1} = P_{c,2} = \cdots = P_{c,N} = P_c$$

where $P_{t,i}$ ($P_{c,i}$) is the probability that the predator will prey upon the ith treated (control) fish next. Parameter K can now be defined by the equation $P_t = K \cdot P_c$. Thus, when a treatment has no effect on prey vulnerability, K=1; when it doubles prey vulnerability, K=2; when it halves prey vulnerability K=0.5, etc.

Since $MP_t + NP_c = 1$ and $P_t = KP_c$, we have $P_t = K/(MK+N)$ and $P_c = 1/(MK+N)$. Note, however, that P_t and P_c change after each successful predatory attack.

The parameter K is closely related to the binomial parameters P and Q. To see this let M=N=R=1; then $P = P_t = \frac{K}{K+1}$ and $Q = P_c = \frac{1}{K+1}$. Recall that a binomial distribution with parameters n, P, and Q is equivalent to the distribution of the sum of N independent Beinoulli trials with parameters P and Q. The reader, however, is cautioned against designing an experiment consisting of Bernoulli trials since there is evidence that learning is involved in the predator-prey process. Although treatment with a stress or toxicant strongly increases prey vulnerability, this effect may not be present until after the first strike.

A maximum likelihood estimator for K is easily obtained by using the program and instructions of Appendix B. The general procedure consists of finding a MLE(K) value such that the likelihood or probability of observing the experimental results for MLE(K) is greater than for any other K value. Algebraically,

$$\text{MLE}(K) = [K: \prod_{i=1}^{L} P(T=T_{i,\text{exp}} | M,N,R_i,K) \geq \prod_{i=1}^{L} P(T=T_{i,\text{exp}} | M,N,R_i,K')$$

$$\text{for every } K' > 0]$$

The statistical procedures which follow are based on one-sided hypothesis tests like H_0: $K=K_0$ vs H_1: $K>K_0$ or H_0':$K=K_0$ vs H_1': $K>K_0$. For most predator-prey studies K_0 will equal one and the alternative hypothesis will be H_1: $K>1$. Nonetheless, the methods and programs that follow are versatile enough to handle a wide range of K_0 values and both one-sided and two-sided hypothesis tests.

Consider an experiment where a treatment A is being examined for its effect on prey vulnerability. In each of L tanks, M treated and N control fish have been placed. The experiment is performed with results (T_i,R_i) for i=1, ... , L. Suppose L equals one; then $P_1 = P(T \geq T_1 | R_1,M,N,K_0)$ is the Type I error for testing H_0: $K=K_0$ versus H_1: $K>K_0$. Thus, P_1 by direct comparison with the α-level for the above hypothesis determines whether we regard K_0 as plausible or conclude K_0 is too low to account for the experimental results. For example, in an experiment where $M=N=R_1=20$ and $L=1$, a result of $T_1=15$ should be regarded as very strong evidence that $K>1.00$ since the $P(T \geq 15|M=N=R_1=20, K=1) = 0.002$.

When L is greater than one, this suggests that some function of P_i values should be used for testing H_0: $K=K_0$ vs H_1: $K>K_0$. The random variable Σ_{K_0} where $\Sigma_{K_0} = \Sigma_{i=1}^{L} P(T \geq T_i|K_0,M,N,R_i) = \Sigma_{i=1}^{L} P_i$ is preferred in this paper since it gives equal emphasis to each P_i value. Values which Σ_{K_0} may assume are finite and range from zero to L. Once K_0, M, N, R_1, ... , R_L are given the true K value completely determines the probabilities assigned to values which Σ_{K_0} may assume; therefore we may make inferences about K by examining the experimental result Σ_{K_0}, exp. A computer program which calculates the Σ_{K_0} distribution for arbitrary K values is provided in Appendix C.

Recall the hypothesis test H_0: $K=K_0$ vs H_1: $K>K_0$. The basic idea behind using the statistic Σ_{K_0} is that the experimental result $\Sigma_{K_0,exp}$ will tend to be improbably low when K is substantially greater than K_0. Suppose an experimenter intended to conduct a predator-prey experiment where $M=N=R_1=R_2=R_3=20$, but was not sure whether the treatment would increase prey vulnerability. Then $K_0=1$ and:

$$P(\Sigma_1 \leq 0.92|K=1) = 0.043$$

$$P(\Sigma_1 \leq 0.92|K=1.5) = 0.415$$

$$P(\Sigma_1 \leq 0.92|K=2.0) = 0.795$$

If the investigator knew the treatment would at least double prey vulnerability, $K_0=2.0$ would be hypothesized and:

$P(\Sigma_2 \leq 0.94|K=2) = 0.041$

$P(\Sigma_2 \leq 0.94|K=3) = 0.375$

$P(\Sigma_2 \leq 0.94|K=4) = 0.739$

These probabilities indicate that H_0: $K=K_0$ will be rejected roughly 75% of the time when the true K value is twice the hypothesized K value.

There are three possible hypothesis tests for predator-prey studies, and a different method applies to each. Test 1: H_0: $K=K_0$ vs H_1: $K>K_0$ is most common. If $P(\Sigma_{K_0} \leq \Sigma_{K_0,exp}\ K=K_0,M,N,R_1,\ldots,R_L) \leq \alpha$, reject H_0; otherwise accept H_0. Test 2 is H_0: $K=K_0$ vs H_1: $K<K_0$. If $P(\Sigma'_{K_0} \leq \Sigma'_{K_0,exp}|K=K_0,M,N,R_1,\ldots,R_L) \leq \alpha$, reject H_0; otherwise accept H_0 (Σ' will be defined later). For test 3: H_0: $K=K_0$ vs H_1: $K=K_0$, reject when $P(\Sigma_{K_0} \leq \Sigma_{K_0,exp}|K=K_0,M,N,R_1,\ldots,R_L) \leq \alpha/2$ or when $P(\Sigma_{K_0} \leq \Sigma_{K_0,exp}|K=K_0,M,N,R_1,\ldots,R_L) \leq \alpha/2$; otherwise accept H_0. Appendices A and C describe how to conduct these tests utilizing a computer.

The statistic Σ_{K_0} also provides a lower confidence limit for K. If the one-sided test H_0: $K=K_0$ vs. H_1: $K>K_0$ is appropriate, find:

$$K_L = \max[K: P(\Sigma_{K_0} \leq \Sigma_{K_0,exp}|K,M,N,R_1,\ldots,R_L) \leq \alpha]$$

For a two-sided test of H_0: $K=K_0$ at level α, find:

$$K_L = \max[K: P(\Sigma_{K_0} \leq \Sigma_{K_0,exp}|K,M,N,R_1,\ldots,R_1) \leq \alpha/2]$$

The practical implementation of these procedures is described in Appendices A and C.

Obtaining upper confidence limits for K requires usage of the new random variable Σ'_{K_0} where $\Sigma'_{K_0} = \Sigma_{i=1}^{L} P(T \leq T_i|K_0,M,N,R_i)$. If the one-sided test H_0: $K=K_0$ vs H_1: $K<K_0$ is appropriate, then:

$$K_U = \min[K: P(\Sigma'_{K_0} \leq \Sigma'_{K_0,exp}|K,M,N,R_1,\ldots,R_L) \leq \alpha]$$

For a two-sided test of H_0: $K=K_0$ at level α, find:

$$K_U = \min[K: P(\Sigma_{K_0} \leq \Sigma_{K_0,exp}|K,M,N,R_1,\ldots,R_L) \leq \alpha/2]$$

In the Sullivan experiment a one-sided test of H_0: $K=1$ vs H_1: $K>1$ at level $\alpha = 0.05$ seemed proper; therefore, to get a confidence interval for K consistent with the hypothesis test, two-sided methods with $\alpha = 0.10$ were used. The Σ and Σ' methods are summarized in Table 8.

For some experimental designs Σ and Σ' methods for finding confidence intervals and testing hypotheses are too costly. In an experiment where M=N=20 and R values are kept close to 20, determining confidence limits for L = 3 on the Purdue CDC processor costs roughly \$1.00 per treatment. For L = 4, the cost would approach \$20.00 per treatment. A very rough formula for cost per treatment is: COST = $M^L/8000$ dollars.

Fortunately, when Σ and Σ' methods become too expensive, maximum order statistics of P_i and P_i' can be used. Let $P_{(1)}$ and $P'_{(1)}$ denote these maximum order statistics. Then the following relations express the essential ideas upon which the $P_{(1)}$ and $P'_{(1)}$ methods are based:

(1) $P_{(1),K} = \max_{i=1,\ldots,L} P(T \geq T_i|K,M,N,R_i)$

(2) $P'_{(1),K} = \max_{i=1,\ldots,L} P(T \leq T_i|K,M,N,R_i)$

TABLE 8

Summary of Σ and Σ' tests.

H_0	H_1	Test Rule
$K=K_0$	$K>K_0$	Reject H_0 if $P(\Sigma_{K_0} \leq \Sigma_{K_0,\exp} \mid K=K_0,M,N,R_1,\ldots,R_L) \leq \alpha$. Otherwise, accept H_0.
$K=K_0$	$K<K_0$	Reject H_0 if $P(\Sigma'_{K_0} \leq \Sigma'_{K_0,\exp} \mid K=K_0,M,N,R_1,\ldots,R_L) \leq \alpha$. Otherwise, accept H_0.
$K=K_0$	$K=K_0$	Reject H_0 if $P(\Sigma_{K_0} \leq \Sigma_{K_0,\exp} \mid K=K_0,M,N,R_1,\ldots,R_L) \leq \alpha/2$ or $P(\Sigma'_{K_0} \leq \Sigma'_{K_0,\exp} \mid K=K_0,M,N,R_1,\ldots,R_L) \leq \alpha/2$. Otherwise, accept H_0.

Two-sided confidence interval (K_L,K_U):

$$K_L = \max\ [K: P(\Sigma_{K_0} \leq \Sigma_{K_0,\exp} \mid K,M,N,R_1,\ldots,R_L) \leq \alpha/2]$$

$$K_U = \min\ [K: P(\Sigma'_{K_0} \leq \Sigma'_{K_0,\exp} \mid K,M,N,R_1,\ldots,R_L) \leq \alpha/2]$$

One-sided confidence interval (K_L,∞):

$$K_L = \max[K: P(\Sigma_{K_0} \leq \Sigma_{K_0,\exp} \mid K,M,N,R_1,\ldots,R_L) \leq \alpha]$$

One-sided confidence interval $(0,K_U)$:

$$K_U = \min[K: P(\Sigma'_{K_0} \leq \Sigma'_{K_0,\exp} \mid K,M,N,R_1,\ldots,R_L) \leq \alpha]$$

A one-sided confidence interval of the form (K_L,∞) can be used if H_0: $K=K_0$ vs H_1: $K>K_0$ is the appropriate hypothesis test and if the investigator is not interested in setting an upper limit for K.

(3) $P(P_{(1),K} \leq P_{(1),K,exp})$

$= P(P_{1,K} \leq P_{(1),K,exp};\ P_{2,K} \leq P_{(1),K,exp};\ldots;P_{L,K} \leq P_{(1),K,exp})$

$= P(P_{1,K} \leq P_{(1),K,exp})P(P_{2,K} \leq P_{(1),K,exp})\ldots P(P_{L,K} \leq P_{(1),K,exp})$

$\leq P_{(1),K,exp}$**L (Fortran notation)

(4) $P_{(1),K,exp}$**L = $\max_{i=1,\ldots,L}$ ($P_{i,K,exp}$**L)

Since $P_{(1)}$ and $P'_{(1)}$ methods are included primarily for completeness, the test methods are presented in Table 9 without further explanation. Appendix D provedes a computer program and practical procedures regarding $P_{(1)}$ and $P'_{(1)}$ tests.

TABLE 9. $P_{(1)}$ and $P'_{(1)}$ Tests

Hypothesis Testing:

H_0:	H_1:	Test Rule
$K=K_0$	$K>K_0$	Reject H_0 if $P_{(1),K_0}$**L $\leq \alpha$; otherwise accept H_0.
$K=K_0$	$K<K_0$	Reject H_0 if $P'_{(1),K_0}$**L $\leq \alpha$; otherwise accept H_0.
$K=K_0$	$K=K_0$	Reject H_0 if $P_{(1),K_0}$**L $\leq \alpha/2$ or $P'_{(1),K_0}$**L $\leq \alpha/2$.

Two Sided Confidence Interval (K_L, K_U):

$$K_L = \max[K: \max_{i=1\ldots L}(P\{T \geq Ti|K,M,N,R_i\}^{**}L) \leq \alpha/2]$$

$$K_U = \min[K: \max_{i=1\ldots L}(P\{T \leq Ti|K,M,N,R_i\}^{**}L) \leq \alpha/2]$$

One Sided Confidence Interval (K_L, ∞):

$$K_L = \max[K: \max_{i=1\ldots L}(P\{T \geq Ti|K,M,N,R_i\}^{**}L) \leq \alpha]$$

One sided Confidence Interval $(0, K_U)$:

$$K_U = \min[K: \max_{i=1\ldots L}(P\{T \leq Ti|K,M,N,R_i\}^{**}L) \leq \alpha]$$

ACKNOWLEDGMENTS

Special thanks go to Virgil L. Anderson for his help and encouragement during and before the writing of this paper. We are grateful to Ashok K. Singh for proofreading the Special Analysis section and to Diana L. Davis for typing this paper. Appreciation is also expressed to Shanti S. Gupta for allowing this non-technical paper to be included in the Mimeograph Series.

REFERENCES

Anderson, V. L. and McLean, R. A. (1974). Design of Experiments. Marcel Decker, NY.

Hogg, R. V. and Craig, A. T. (1970). Introduction to Mathematical Statistics. Macmillan, NY.

Appendix A: Program PDE

Program PDE computes the probability density and cumulative distribution functions for random variable T given K, M, N and R. Included in the output are M,N,R,K,E(T/R|M,N,R,K) and Var(T/R|M,N,R,K). The computations rely on the recursion formula:

$$P(T=i|R,M,N,K) = P(T=i|(R-1),M,N,K)P(R^{th}\text{ fish eaten is a control fish})$$
$$+ P(T=i-1|(R-1),M,N,K)P(R^{th}\text{ fish eaten is a treated fish})$$
$$= P(T=i|(R-1),M,N,K)*(N-R+i+1)/[K(M-i)+(N-R+i+1)]$$
$$+ P(T=i-1|(R-1),M,N,K)*K(M-i+1)/[K(M-i+1)+(N-R+i)]$$

The important probabilities, $P_{i,K} = P(T \geq T_i|K,M,N,R_i)$ and $P'_{i,K} = P(T \leq T_i|K,M,N,R_i)$, are easily extracted from the PDE output. Thus, using this program is a first step for Σ and Σ' methods since $\Sigma_{K_0,exp} = \Sigma_{i=1}^{L} P_{i,K_0}$ and $\Sigma'_{K_0,exp} = \Sigma_{i=1}^{L} P'_{i,K_0}$.

Program PDE may require some modification before it can be used on your computer system. The Program PDE card may need to be removed, in which case Tape 5 and Tape 6 must be set as input and output on your control cards. The format statements which are non-Hollerith will cause no problems on most systems. Otherwise, Program PDE is written in standard Fortran. Note: Modifications which must be made for Program PDE will also need to be made for other programs in this paper.

Program input is provided by using the following data deck:

Card	Information	Format
1	RMIN,RMAX,NT,NC,K,KINC,KMAX	(2I2,X,5F5.0)
...	...	...
LAST	Blank	

Fortran variables in the above list have the following meanings:

(1) RMIN is the minimum R value for which a cumulative distribution table will be printed.

(2) RMAX is the maximum R value for which a CDF table will be printed.

(3) NT(NT=M) is the number of treated fish per tank at the beginning of an experiment.

(4) NC(NC=N) is the number of control fish per tank at the beginning of an experiment.

(5) K is an initial value for which a CDF table is desired.

(6) KMAX is the maximum value for which a CDF table is desired.

(7) KINC is the amount K is incremented until KMAX is reached.

Tables are printed for all combinations of (RMIN,RMIN+1,...,RMAX) and (K,K+KINC,K+(2*KINC),...,KMAX).

Example: The experimental results for the Sullivan experiment for 0.025 mg. Cd/liter and 21 days exposure ($M=N=20, K_0=1$) are:

Tank:	1	2	3
R_i	21	20	20
T_i	15	12	17

The data cards should be:

```
Card/Cols:  123456789012345678901234567890
    1       2021 20.0 20.0 1.00 1.00 1.00
    2       A blank card.
```

The resultant P_i and P_i' values are:

Tank:	1	2	3
$P_{i,K_0} = 1$	.005193	.171534	.000010
$P_{i}',_{K_0} = 1$	.999384	.943583	1.000000

<u>Program Limitations</u>:

1. $2 \leq M,N,RMAX \leq 33$

2. $RMAX \leq M+N$

3. $K > 0.001$

4. In the programs that follow the three above restrictions must also be met; otherwise, the programs require modification. In addition, the programs of Appendices B, C and D require that the pair (M,N) for a given treatment remain the same for all L subexperiments. Thus, without serious changes, programs of these latter appendices cannot be used to analyze the 0.05 mg. Cd/1., 48 hr. treatment of Table 1. The reader is therefore advised to avoid such complications by designing experiments with all (M,N) pairs the same for the L subexperiments of each treatment.

An entire program deck will consist of the following:

(1) Jobcard
(2) Password card
(3) Control card(s)
(4) 7/8/9 multi-punch in column one
(5) Program cards
(6) 7/8/9 multi-punch in column one
(7) Data cards
(8) 6/7/8/9 multi-punch in column one

Program PDE follows on the next page.

```
      PROGRAM PDE(INPUT,OUTPUT,TAPE5=INPUT,TAPE6=OUTPUT)
      REAL P(35),REG(35),PDF(35),CDF(35)
      REAL NT,NC,K,KINC,KMAX,FI,R,EP,EPSQ,VARP
      INTEGER  T(35),IM,RMAX,RMIN,R2,LO,HI
1000  READ(5,10) RMIN,RMAX,NT,NC,K,KINC,KMAX
10    FORMAT(2I2,X,5F5.0)
      IF(NT.GT.33.0)  STOP
      IF(NC.GT.33.0)  STOP
      IF(RMAX.GT.33)  STOP
      IF(KINC.LT.0.001)   STOP
      RMIN=RMIN+2
      RMAX=RMAX+2
      HI=IFIX(NT+2.1)
2000  CONTINUE
      DO 55 I=1,35
      REG(I)=0.0
      PDF(I)=0.0
      CDF(I)=0.0
55    CONTINUE
      PDF(2)=NC/(NT*K+NC)
      PDF(3)=NT*K/(NT*K+NC)
      DO 44 L=4,RMAX
      R2=L
      R=FLOAT(R2-2)
      LO=IFIX(R-NC+2.1)
      DO 11 I=2,R2
      FI=FLOAT(I)
      P(I)=(FI-2.0)/R
      T(I)=I-2
      REG(I)=0.0
      IF((I.LT.LO).OR.(I.GT.HI))  GO TO 12
      IM=I-1
      REG(I)=PDF(IM)*((NT-FI+3.0)*K/((NT-FI+3.0)*K+NC-R+FI-2.0))
     *      +PDF(I)*((NC-R+FI-1.0)/((NT-FI+2.0)*K+NC-R+FI-1.0))
12    CONTINUE
11    CONTINUE
      DO 66 I=2,R2
      PDF(I)=REG(I)
66    CONTINUE
      IF(R2.LT.RMIN) GO TO 45
      EP=0.0
      EPSQ=0.0
      DO 22 I=2,R2
      IM=I-1
      CDF(I)=CDF(IM)+PDF(I)
      EP=EP+PDF(I)*P(I)
      EPSQ=EPSQ+PDF(I)*P(I)*P(I)
22    CONTINUE
      VARP=EPSQ-EP*EP
      WRITE(6,33) NT,NC,R,K,EP,VARP,(P(I),PDF(I),CDF(I),T(I),I=2,R2)
33    FORMAT(///5X,≠TOTAL(T)=≠,F5.2,≠  TOTAL(C)=≠,F5.2,≠  R=≠,F5.2,
     *     ≠  K=≠,F6.3,≠  E(P)=≠,F8.6,≠  VAR(P)=≠,F8.6//5X,
     *     ≠N(T)/N(R)≠,7X,≠P(T.EQ.TEXP)≠,4X,≠P(T.LE.TEXP)≠,4X,≠TEXP≠//
     *     (5X,F9.6,7X,F9.6,6X,F9.6,7X,I2))
45    CONTINUE
44    CONTINUE
      K=K+KINC
      IF(K.GT.KMAX) GO TO 1000
      GO TO 2000
      END
```

Appendix B: Program MLE

Program MLE computes the likelihood of an experimental result for selected K values. Computer searching techniques can yield a maximum likelihood estimate of K which is accurate to two or three decimal places.

The data deck for this program should be as follows:

Card	Information	Format
1	TANKS(=L value)	(I1)
2	RMIN,RMAX,NT,NC,K,KINC,KMAX	(2I2,X,5F5.0)
3	R_1,T_1	(2I2)
4	R_2,T_2	
...	...	...
L+2	R_L,T_L	
**	Cards 2 thru L+2 form a set which may be repeated if one wishes to conduct more than one search per program run. The TANKS value is listed only once.	
Last 10	Ten blank data cards.	

Explanation:

(1) RMIN is set to $\min_{i=1,\ldots,L} R_i$

(2) RMAX is set to $\max_{i=1,\ldots,L} R_i$

(3) K,KINC,KMAX define K values for which the likelihood of the experimental result will be listed.

Example: The first step in getting a MLE of K is to guess roughly where the MLE(K) will be. A fair initial guess is $G = \Sigma_{i=1}^{L} T_i / \Sigma_{i=1}^{L} C_i$, which equals 2.59 in our example. We therefore search K = 2.5,3.0,...,5.0 using the following data deck:

```
Card/Cols:  123456789012345678901234567890
  1         3
  2         2021 20.0 20.0 2.50 0.50 5.00
  3         2115
  4         2012
  5         2017
 6-15       Blank cards
```

We get output:

```
K = 2.500     Likelihood = .000312
K = 3.000     Likelihood = .000708
K = 3.500     Likelihood = .001019
K = 4.000     Likelihood = .001109
K = 4.500     Likelihood = .001008
K = 5.000     Likelihood = .000815
```

Since the zenith is near 4.00, we run K = 3.80,3.85,...,4.20. The result is MLE(K) = 3.95.

Program MLE follows on the next page.

```
      PROGRAM MLE(INPUT,OUTPUT,TAPE5=INPUT,TAPE6=OUTPUT)
      REAL P(35),REG(35),PDF(35),CDF(35)
      REAL NT,NC,K,KINC,KMAX,FI,R,EP,EPSQ,VARP
      INTEGER  T(35),IM,RMAX,RMIN,R2,LO,HI
      REAL  LIKE(3),LIKELY
      INTEGER TANKS,RR(3),TT(3),RM,IR,TTJ2
      READ(5,40) TANKS
40    FORMAT(I1)
1000  READ(5,10) RMIN,RMAX,NT,NC,K,KINC,KMAX
10    FORMAT(2I2,X,5F5.0)
      IF(NT.GT.33.0)  STOP
      IF(NC.GT.33.0)  STOP
      IF(RMAX.GT.33)  STOP
      IF(KINC.LT.0.001)   STOP
      READ(5,50) (RR(J),TT(J),J=1,TANKS)
50    FORMAT(2I2)
      DO 77 J1=1,TANKS
      TT(J1)=TT(J1)+2
77    CONTINUE
      RM=RMAX
      RMIN=RMIN+2
      RMAX=RMAX+2
      HI=IFIX(NT+2.1)
2000  CONTINUE
      DO 55 I=1,35
      REG(I)=0.0
      PDF(I)=0.0
      CDF(I)=0.0
55    CONTINUE
      PDF(2)=NC/(NT*K+NC)
      PDF(3)=NT*K/(NT*K+NC)
      DO 44 L=4,RMAX
      R2=L
      R=FLOAT(R2-2)
      LO=IFIX(R-NC+2.1)
      DO 11 I=2,R2
      FI=FLOAT(I)
      P(I)=(FI-2.0)/R
      T(I)=I-2
      REG(I)=0.0
      IF((I.LT.LO).OR.(I.GT.HI))  GO TO 12
      IM=I-1
      REG(I)=PDF(IM)*((NT-FI+3.0)*K/((NT-FI+3.0)*K+NC-R+FI-2.0))
     *       +PDF(I)*((NC-R+FI-1.0)/((NT-FI+2.0)*K+NC-R+FI-1.0))
12    CONTINUE
11    CONTINUE
      DO 66 I=2,R2
      PDF(I)=REG(I)
66    CONTINUE
      IF(R2.LT.RMIN)  GO TO 45
      IR=IFIX(R+0.1)
      DO 88 J2=1,TANKS
```

```
      TTJ2=TT(J2)
      IF(IR.EQ.RR(J2))  LIKE(J2)=PDF(TTJ2)
88    CONTINUE
      IF(IR.EQ.RM)  LIKELY=LIKE(1)*LIKE(2)*LIKE(3)
      IF(IR.EQ.RM)  WRITE(6,30) K,LIKELY
30    FORMAT(5X,≠K=≠,F6.3,5X,≠LIKELIHOOD=≠,F8.6)
45    CONTINUE
44    CONTINUE
      K=K+KINC
      IF(K.GT.KMAX) GO TO 1000
      GO TO 2000
      END
```

Appendix C: Program SIGMAD

Program SIGMAD computes the cumulative distribution functions of Σ_{K_0} and Σ'_{K_0}. It may be used in conjunction with a computer search to find upper and/or lower confidence limits for K.

The data deck for SIGMAD should be as follows:

Card	Information		Format
1	$K_0, \Sigma_{K_0,exp}$ or $\Sigma'_{K_0,exp}$ (whichever is applicable)		(F5.0,F10.0)
2	Tanks,RMAX,NT,NC,K_0,1.00,K_0		(I1,I2,5F5.0)
3	SMIN,SMAX		(2F5.0)
4	R_1		(I2)
5	R_2		(I2)
...	...		...
L+3	R_L		(I2)
L+4	Tanks,RMAX,NT,NC,K,KINC,KMAX		(I2,I2,5F5.0)
L+5	SMIN,SMAX	Repeat of cards 3 thru L+3	(2F5.0)
L+6	R_1		(I2)
L+7	R_2		(I2)
...	...		...
2L+5	R_L		(I2)
LAST 10	Ten blank cards		

Explanation:

(1) K_L and K_N for a given treatment must be sought after one at a time. When seeking K_L, the $\Sigma_{K_0,exp}$ value is listed on card 1. List $\Sigma'_{K_0,exp}$ on card 1 when searching for K_U.

(2) RMAX = $\max_{i=1,\ldots,L} R_i$

(3) SMIN and SMAX specify the values of Σ_{K_0} or Σ'_{K_0} for which the CDF values are desired. For example, setting SMIN = .58 and SMAX = .59 when $\Sigma_{K_0,exp}$ = .586 will cause $P(\Sigma_{K_0} \le .586|K,M,N,R_1,\ldots,R_L)$ and $P(\Sigma'_{K_0} \le .586|K,M,N,R_1,\ldots,R_L)$ to be printed.

Example (cont'd): From Appendix A we know $\Sigma_{K_0=1,exp} = \Sigma_{i=1}^{3} P_{i,K_0=1}$ $= .176737$ and that $\Sigma'_{K_0=1,exp} = \Sigma_{i=1}^{3} P'_{i,K_0=1} = 2.942967$. Thus, for a two-sided 90% confidence interval,

$$K_L = \max[K: P(\Sigma_{K_0} \leq 0.17674 | M=N=20, R_1=21, R_2=R_3=20, K) \leq .05]$$

and

$$K_U = \min[K: P(\Sigma'_{K_0} \leq 2.94297 | M=N=20, R_1=21, R_2=R_3=20, K) \leq .05]$$

The first step in finding K_L and K_U is to guess at a wide region for their location. For K_L we search at K = 1.0,1.5,...,3.5; for K_U we search at K = 5.0,6.0,10.0. The correct data decks are (A) the K_L deck:

```
Card/Cols:  1234567890123456789012345678
 1          1.00 0.176737
 2          321 20.0 20.0 1.00 1.00 1.00
 3          0.17 0.18
 4          21
 5          20
 6          20
 7          321 20.0 20.0 1.00 0.50 3.50
 8          0.17 0.18
 9          21
10          20
11          20
12-21       Ten blank cards
```

(B) the K_U deck:

```
Card/Cols:  123456789012345678901234567890
 1          1.00 2.942967
 2          321 20.0 20.0 1.00 1.00 1.00
 3          2.94 2.95
 4          21
 5          20
 6          20
 7          321 20.0 20.0 5.00 1.00 10.0
 8          2.94 2.95
 9          21
10          20
11          20
12-21       Ten blank cards
```

These two decks must be run separately.

The output reveals that K_L is between 1.5 and 2.0 since $P(\Sigma_{K_0} \leq .1767|K=1.5) = .0169$ and $P(\Sigma_{K_0} \leq .1767|K-2.0) = .1252$; it also shows that K_U is between 5.0 and 6.0 since $P(\Sigma'_{K_0} \leq 2.9430|K=5) = .0574$ and $P(\Sigma'_{K_0} \leq 2.9430|K=6) = .0201$. Therefore we continue our search by replacing data cards A7 and B7 by cards A and B, respectively:

```
Cards/Cols:  12345678901234567890123456789
  A          321 20.0 20.0 1.60 0.10 1.90
  B          321 20.0 20.0 5.10 0.10 5.90
```

After reviewing the output, we conclude that $(K_L,K_U) \cong (1.7,5.2)$ since the $P(\Sigma_{K_0} \leq .1767|K=1.7) = .0452$ and $P(\Sigma'_{K_0} \leq 2.9430|K=5.2) = .0464$.

Bonuses from the output of data decks A and B are:

(1) $P(\Sigma_{K_0} \leq 0.1767|M=N=20,R_1=21,R_2=R_3=20,K=K_0=1) = .0002$.

Since this probability is less than .05, we reject H_0: K=1 in favor of H_1:K > 1.

(2) $P(\Sigma'_{K_0} \leq 2.9430|M=N=20,R_1=21,R_2=R_3=20,K=K_0=1) = .9996$.

If we were testing H_0: K=1 vs. H_1: K < 1, we would accept the null hypothesis.

The Σ and Σ' methods presented here are designed to ensure that 100(1-α)% confidence intervals are their given size or larger. This has been done by making the Fortran variable ADJUST 0.01 smaller than it should be if the computer calculations were arithmetically precise. But this protective measure can cause the confidence intervals to become overly large when $\Sigma_{K_0,exp}$ or $\Sigma'_{K_0,exp}$ are near zero or L. This problem, however, can be overcome by using the MLE(K) in place of K_0. Simply use

Program PDE to compute $\Sigma_{MLE(K),exp}$ and $\Sigma'_{MLE(K),exp}$; then find $K_L = [K: P(\Sigma_{MLE(K)} \leq \Sigma_{MLE(K),exp} | K,M,N,R_1,\ldots,R_L) \leq \alpha/2]$ and $K_U = [K: P(\Sigma'_{MLE(K)} \leq \Sigma'_{MLE(K),exp} | K,M,N,R_1,\ldots,R_L) \leq \alpha/2]$ using Program SIGMAD.

At this point the Σ and Σ' methods offer an infinite number of procedures (one for every $K_0 > 0$) for finding a confidence interval for K. This represents an uncomfortable situation since each procedure may yield somewhat different confidence intervals. Originally no such problem existed because K_0 was postulated to equal one in all predator-prey experiments. When the authors decided to allow K_0 to assume any value an experimenter believed to be appropriate, the problem of multiple confidence intervals arose. To resolve the multiple interval and computer accuracy problems simultaneously, the authors suggest that experimenters use exclusively the $\Sigma_{MLE(K)}$ and $\Sigma'_{MLE(K)}$ methods for calculating confidence intervals. The Σ_{K_0} and Σ'_{K_0} methods must still be used to get hypothesis test P values.

Program SIGMAD, as presented, works for experiments where the number of tanks per treatment is three. When the number of tanks is two or four (five tanks probably makes this program too expensive to use), amend this program as follows:

(a) For L=TANKS=2, (1) replace the 3 in 3I4 of format statement 40 with a 2; (2) replace all statements between "DO 1 ... Big" and "1 Continue" by:

```
      DO 1 I1 = Small, Big
      DO 2 I2 = Small, Big
      Sump = Ifix(100.0*(FF(I1,1) + FF(I2,2)) + Adjust)
      Pdist(Sump) = Pdist(Sump) + (PP(I1,1)*PP(I2,2))
      Spofg = Ifix(100.0*(GG(I1,1)+GG(I2,2)) + Adjust)
      Pdofg(Spofg) = Pdofg(Spofg)+(PP(I1,1)*PP(I2,2))
    2 Continue
    1 Continue
```

(b) For L=TANKS=4, (1) replace the 3 in 3I4 of format statement 40 with a 4; (2) replace all statements between "DO 1 ... Big" and "1 Continue" by:

```
      DO 1 I1= Small, Big
      DO 2 I2= Small, Big
      DO 3 I3= Small, Big
      DO 4 I4= Small, Big
      Sump = Ifix (100.0*[FF(I1,1) + FF(I2,2) + FF(I3,3) + FF(I4,4)] + ADJUST)
      Pdist(Sump) = Pdist(Sump) + (PP(I1,1)*PP(I2,2)*PP(I3,3)*PP(I4,4))
      Spofg = Ifix(100.0*[GG(I1,1) + GG(I2,2) + GG(I3.3) + GG(I4.4)] + ADJUST)
      Pdofg(Spofg) = Pdofg(Spofg) + (PP(I1,1)*PP(I2,2)*PP(I3.3)*PP(I4,4))
    4 Continue
    3 Continue
    2 Continue
    1 Continue
```

Program SIGMAD starts on the next page.

```
      PROGRAM SIGMAD(INPUT,OUTPUT,TAPE5=INPUT,TAPE6=OUTPUT)
      REAL  ADJUST,EXCESS,SIGEXP
      REAL  KZERO,KZEROL,KZEROH
      REAL  PP(35,5),FF(35,5),GG(35,5)
      REAL  PDIST(502),CDIST(502),V(502)
      REAL  NT,NC,K,KINC,KMAX,FI,R
      REAL  P(35),REG(35),PDF(35),CDF(35),G(35),CDOFG(502),PDOFG(502)
      INTEGER  RS(5),TANKS,LINDX
      INTEGER  SPOFG,SUMP,RR(5),RMAX,R2,IM,L,LO,HI
      INTEGER  BIG,SMALL,RMAX2
      READ(5,60)   KZERO,SIGEXP
60    FORMAT(F5.0,F10.0)
      EXCESS=((100.0*SIGEXP)-FLOAT(IFIX(100.0*SIGEXP)))*0.01
      ADJUST=1.99-(100.0*EXCESS)
      KZEROL=KZERO-0.0001
      KZEROH=KZERO+0.0001
1000  READ(5,10)   TANKS,RMAX,NT,NC,K,KINC,KMAX
10    FORMAT(I1,I2,5F5.0)
      IF(NT.GT.33.0)   STOP
      IF(NC.GT.33.0)   STOP
      IF(RMAX.GT.33)   STOP
      IF(KINC.LT.0.001)    STOP
      L=TANKS
      LINDX=(100*L)+1
      RMAX=RMAX+2
      HI=IFIX(NT+2.1)
      READ(5,797) SMIN,SMAX
797   FORMAT(2F5.0)
      DO 11 I=1,L
      READ(5,20)   RS(I)
20    FORMAT(I2)
      RR(I)=RS(I)+2
11    CONTINUE
2000  CONTINUE
      DO 22 I=1,35
      REG(I)=0.0
      PDF(I)=0.0
      CDF(I)=0.0
22    CONTINUE
      PDF(2)=NC/(NT*K+NC)
      PDF(3)=(NT*K)/(NT*K+NC)
      DO 33 LL=4,RMAX
      R2=LL
      R=FLOAT(R2-2)
      LO=IFIX(R-NC+2.1)
      DO 44 I=2,R2
      FI=FLOAT(I)
      P(I)=(FI-2.0)/R
      REG(I)=0.0
      IF((I.LT.LO).OR.(I.GT.HI))   GO TO 12
      IM=I-1
```

```
      REG(I)=PDF(IM)*((NT-FI+3.0)*K/((NT-FI+3.0)*K+NC-R+FI-2.0))
     *      +PDF(I)*((NC-R+FI-1.0)/((NT-FI+2.0)*K+NC-R+FI-1.0))
12    CONTINUE
44    CONTINUE
      DO 55 I=2,R2
      PDF(I)=REG(I)
55    CONTINUE
      DO 780 JJ=1,L
      IF(R2.EQ.RR(JJ))  GO TO 35
780   CONTINUE
      GO TO 34
35    CONTINUE
      DO 66 I=2,R2
      IM=I-1
      CDF(I)=CDF(IM)+PDF(I)
      G(I)=1.0-CDF(IM)
66    CONTINUE
      DO 111 JA=1,L
      IF(R2.NE.RR(JA))  GO TO 333
      DO 222 JB=2,R2
      PP(JB,JA)=PDF(JB)
      IF(K.LT.KZEROL.OR.K.GT.KZEROH)  GO TO 444
      FF(JB,JA)=CDF(JB)
      GG(JB,JA)=G(JB)
444   CONTINUE
222   CONTINUE
333   CONTINUE
111   CONTINUE
34    CONTINUE
33    CONTINUE
      DO 77 N=1,502
      PDIST(N)=0.0
      PDOFG(N)=0.0
77    CONTINUE
      DO 2222 J=2,RMAX
      SMALL=J
      DO 6666 JD=1,L
      IF(PP(J,JD).GT.0.0001)  GO TO 3333
6666  CONTINUE
2222  CONTINUE
3333  CONTINUE
      RMAX2=RMAX+2
      DO 4444 J=2,RMAX2
      BIG=RMAX2-J
      DO 7777 JE=1,L
      IF(PP(BIG,JE).GT.0.0001)  GO TO 5555
7777  CONTINUE
4444  CONTINUE
5555  CONTINUE
      DO 1 I1=SMALL,BIG
      DO 2 I2=SMALL,BIG
      DO 3 I3=SMALL,BIG
```

```
      SUMP=IFIX(100.0*(FF(I1,1)+FF(I2,2)+FF(I3,3))+ADJUST)
      PDIST(SUMP)=PDIST(SUMP)+(PP(I1,1)*PP(I2,2)*PP(I3,3))
      SPOFG=IFIX(100.0*(GG(I1,1)+GG(I2,2)+GG(I3,3))+ADJUST)
      PDOFG(SPOFG)=PDOFG(SPOFG)+(PP(I1,1)*PP(I2,2)*PP(I3,3))
3     CONTINUE
2     CONTINUE
1     CONTINUE
      CDIST(1)=PDIST(1)
      CDOFG(1)=PDOFG(1)
      V(1)=EXCESS
      DO 74 I=2,LINDX
      IM=I-1
      CDIST(I)=CDIST(IM)+PDIST(I)
      CDOFG(I)=CDOFG(IM)+PDOFG(I)
      V(I)=FLOAT(IM)*0.01+EXCESS
74    CONTINUE
      WRITE(6,40)  K,(RS(I),I=1,L),KZERO
40    FORMAT(//5X,≠SIGMA+SIGMAPRIME P(SIGMAP.LE.SIGMAPEXP) ≠,
     *       ≠P(SIGMA.LE.SIGMAEXP)  K=≠,F6.3,3I4,≠  KZERO=≠,F6.3)
      DO 75 I=1,LINDX
      IF(V(I).GT.SMIN.AND.V(I).LT.SMAX)
     *   WRITE(6,50) V(I),CDIST(I),CDOFG(I)
50    FORMAT(10X,F6.4,10X,F9.4,10X,F9.4)
75    CONTINUE
      K=K+KINC
      IF(K.GT.KMAX) GO TO 1000
      GO TO 2000
      END
```

Appendix D: Program P1TOL

Program P1TOL computes $P_{i,K}$**L and $P'_{i,K}$**L values where $P_{i,K} = P(T \geq T_i | K,M,N,R_i)$ and $P'_{i,K} = P(T \leq T_i | K,M,N,R_i)$. Used in conjunction with Table 9, this program will aid in finding confidence limits for K when SIGMAD is too expensive to use.

The data deck for P1TOL must contain the following information:

Card	Information	Format
1	Tanks per treatment	(I1)
2	RMIN,RMAX,NT,NC,K,KINC,KMAX	(2I2,X,5F5.0)
3	R_1,T_1	(2I2)
4	R_2,T_2	(2I2)
...	...	...
L+2	R_1,T_L	(2I2)
...	Searches for (K_L,K_U) may be conducted for more than one treatment, provided the value for TANKS remains the same. Simply follow the format of cards 2 thru L+2.	
LAST 10	Ten blank cards	

Continuing our example we search for K_L below MLE(K) = 3.95 and for K_U above 3.95. Since this program is fairly cheap to run, look at K = 1.0,1.1,...,3.0 for K_L and at K = 5.0,5.5,...,20.0 for K_U. Conclude that the 90% confidence limits for K are K_L = 1.3 and K_U = 18.5 since the $\max_{i=1,2,3}(P_{i,K=1.3}$**3) = .0454,

the $\max_{i=1,2,3}(P_{i,K=1.4}$**3) = 0.738,

the $\max_{i=1,2,3}(P_{i,K=18.0}$**3) = .0527 and

the $\max_{i=1,2,3}(P_{i,K=18.5}$**3) = .0467.

The data deck for this example is:

```
Card/Cols:  123456789012345678901234567890
  1         3
  2         2021 20.0 20.0 1.00 0.10 3.00
  3         2115
  4         2012
  5         2017
  6         2021 20.0 20.0 5.00 0.50 20.0
  7         2115
  8         2012
  9         2017
10-19       Ten blank cards
```

Since $\max_{i=1,2,3}(P_{i,K=1.0}**3) = 0.0050$, we reject H_0: K=1.0 in favor of H_1: K > 1.00.

Program P1TOL follows on the next page.

```
      PROGRAM P1TOL(INPUT,OUTPUT,TAPE5=INPUT,TAPE6=OUTPUT)
      REAL P(35),REG(35),PDF(35),CDF(35)
      REAL NT,NC,K,KINC,KMAX,FI,R,EP,EPSQ,VARP
      INTEGER  T(35),TANKS
      INTEGER IM,RMAX,RMIN,R2,LO,HI
      REAL  PMAXPR,PPMXPR,PVALUE,PPRIME
      INTEGER  TT(3),RR(3),T1,T2,L,IR
      READ(5,50) TANKS
50    FORMAT(I1)
      LL=TANKS
      WRITE(6,30) LL
30    FORMAT(6X,≠  K      R      P**L    PP**L     L=≠,I2//)
1000  READ(5,10) RMIN,RMAX,NT,NC,K,KINC,KMAX
10    FORMAT(2I2,X,5F5.0)
      IF(NT.GT.33.0)  STOP
      IF(NC.GT.33.0)  STOP
      IF(RMAX.GT.33)  STOP
      IF(KINC.LT.0.001)   STOP
      RMIN=RMIN+2
      RMAX=RMAX+2
      HI=IFIX(NT+2.1)
      READ(5,40)  (RR(I),TT(I),I=1,LL)
40    FORMAT(2I2)
2000  CONTINUE
      DO 55 I=1,35
      REG(I)=0.0
      PDF(I)=0.0
      CDF(I)=0.0
55    CONTINUE
      PDF(2)=NC/(NT*K+NC)
      PDF(3)=NT*K/(NT*K+NC)
      DO 44 L=4,RMAX
      R2=L
      R=FLOAT(R2-2)
      LO=IFIX(R-NC+2.1)
      DO 11 I=2,R2
      FI=FLOAT(I)
      P(I)=(FI-2.0)/R
      T(I)=I-2
      REG(I)=0.0
      IF((I.LT.LO).OR.(I.GT.HI))   GO TO 12
      IM=I-1
      REG(I)=PDF(IM)*((NT-FI+3.0)*K/((NT-FI+3.0)*K+NC-R+FI-2.0))
     *        +PDF(I)*((NC-R+FI-1.0)/((NT-FI+2.0)*K+NC-R+FI-1.0))
12    CONTINUE
11    CONTINUE
      DO 66 I=2,R2
      PDF(I)=REG(I)
66    CONTINUE
      IF(R2.LT.RMIN)   GO TO 45
      EP=0.0
      EPSQ=0.0
```

```
      DO 22 I=2,R2
      IM=I-1
      CDF(I)=CDF(IM)+PDF(I)
      EP=EP+PDF(I)*P(I)
      EPSQ=EPSQ+PDF(I)*P(I)*P(I)
22    CONTINUE
      PMAXPR=0.0
      PPMXPR=0.0
      IR=IFIX(R+0.1)
      DO 77 I=1,LL
      T1=TT(I)+1
      T2=TT(I)+2
      IF(IR.EQ.RR(I)) PVALUE=1.00-CDF(T1)
      IF(IR.EQ.RR(I)) PPRIME=CDF(T2)
      PMAXPR=PVALUE**LL
      PPMXPR=PPRIME**LL
      IF(IR.EQ.RR(I)) WRITE(6,20) K,IR,PMAXPR,PPMXPR
77    CONTINUE
20    FORMAT(5X,F6.3,I5,F8.4,F9.4)
45    CONTINUE
44    CONTINUE
      K=K+KINC
      IF(K.GT.KMAX) GO TO 1000
      GO TO 2000
      END
```